Dror Bar-Natan: Classes: 2014-15: Math 240 Algebra I: Wednesday December 3:

$$\left\{\begin{pmatrix} 4 & 3 \\ -6 & -5 \end{pmatrix}.\begin{pmatrix} 4 & 3 \\ -6 & -5 \end{pmatrix}.\begin{pmatrix} 4 & 3 \\ -6 & -5 \end{pmatrix}.\begin{pmatrix} 4 & 3 \\ -6 & -5 \end{pmatrix}.\begin{pmatrix} 4 & 3 \\ -6 & -5 \end{pmatrix} \text{ // MatrixForm}, \begin{pmatrix} 1 & -1 \\ -1 & 2 \end{pmatrix}.\begin{pmatrix} 1 & 0 \\ 0 & -32 \end{pmatrix}.\begin{pmatrix} 2 & 1 \\ 1 & 1 \end{pmatrix} \text{ // MatrixForm}\right\}$$

$$\left\{\begin{pmatrix} 34 & 33 \\ -66 & -65 \end{pmatrix}, \begin{pmatrix} 34 & 33 \\ -66 & -65 \end{pmatrix}\right\}$$

```
F[0] = F[1] = 1;
F[n_] /; n > 1 := F[n] = F[n - 1] + F[n - 2];
F /@ Range[0, 10]
```

{1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89}

$$\text{A} = \begin{pmatrix} 0 & 1 \\ 1 & 1 \end{pmatrix};$$

```
MatrixPower[A, 10] // MatrixForm
```

$$\begin{pmatrix} 34 & 55 \\ 55 & 89 \end{pmatrix}$$

```
{MatrixPower[A, 50][[2, 2]], F[50]}
```

{20 365 011 074, 20 365 011 074}

```
(B = A - λ IdentityMatrix[2]) // MatrixForm
```

$$\begin{pmatrix} -\lambda & 1 \\ 1 & 1-\lambda \end{pmatrix}$$

```
χ = Det[B]
```

$-1 - \lambda + \lambda^2$

```
{λ1, λ2} = λ /. Solve[χ == 0, λ]
```

$$\left\{\frac{1}{2}\left(1-\sqrt{5}\right), \frac{1}{2}\left(1+\sqrt{5}\right)\right\}$$

```
(DD = DiagonalMatrix[{λ1, λ2}]) // MatrixForm
```

$$\begin{pmatrix} \frac{1}{2}\left(1-\sqrt{5}\right) & 0 \\ 0 & \frac{1}{2}\left(1+\sqrt{5}\right) \end{pmatrix}$$

```
{v1} = NullSpace[B /. λ → λ1]
```

$$\left\{\left\{\frac{1}{2}\left(-1-\sqrt{5}\right), 1\right\}\right\}$$

```
{v2} = NullSpace[B /. λ → λ2]
```

$$\left\{\left\{\frac{1}{2}\left(-1+\sqrt{5}\right), 1\right\}\right\}$$

```
(CC = Transpose[{v1, v2}]) // MatrixForm
```

$$\begin{pmatrix} \frac{1}{2}\left(-1-\sqrt{5}\right) & \frac{1}{2}\left(-1+\sqrt{5}\right) \\ 1 & 1 \end{pmatrix}$$

```
(CCinv = Inverse[CC]) // MatrixForm
```

$$\begin{pmatrix} -\frac{1}{\sqrt{5}} & -\frac{1-\sqrt{5}}{2\sqrt{5}} \\ \frac{1}{\sqrt{5}} & -\frac{-1-\sqrt{5}}{2\sqrt{5}} \end{pmatrix}$$

```
CCinv.CC // Simplify // MatrixForm
```

$$\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$$

```
CC.DD.CCinv // Simplify // MatrixForm
```

$$\begin{pmatrix} 0 & 1 \\ 1 & 1 \end{pmatrix}$$

```
(DDn = DiagonalMatrix[{λ1^n, λ2^n}]) // MatrixForm
```

$$\begin{pmatrix} \left(\frac{1}{2}\left(1-\sqrt{5}\right)\right)^n & 0 \\ 0 & \left(\frac{1}{2}\left(1+\sqrt{5}\right)\right)^n \end{pmatrix}$$

```
CC.DDn.CCinv // Simplify // MatrixForm
```

$$\begin{pmatrix} \frac{2^{-1-n}\left(\left(1-\sqrt{5}\right)^n\left(1+\sqrt{5}\right)+\left(-1+\sqrt{5}\right)\left(1+\sqrt{5}\right)^n\right)}{\sqrt{5}} & \frac{2^{-n}\left(-\left(1-\sqrt{5}\right)^n+\left(1+\sqrt{5}\right)^n\right)}{\sqrt{5}} \\ \frac{-\left(\frac{1}{2}\left(1-\sqrt{5}\right)\right)^n+\left(\frac{1}{2}\left(1+\sqrt{5}\right)\right)^n}{\sqrt{5}} & \frac{2^{-1-n}\left(-\left(1-\sqrt{5}\right)^{1+n}+\left(1+\sqrt{5}\right)^{1+n}\right)}{\sqrt{5}} \end{pmatrix}$$

```
Formula = (CC.DDn.CCinv)[[2, 2]] // Simplify
```

$$\frac{2^{-1-n}\left(-\left(1-\sqrt{5}\right)^{1+n}+\left(1+\sqrt{5}\right)^{1+n}\right)}{\sqrt{5}}$$

```
Formula /. n → 50
```

$$\frac{-\left(1-\sqrt{5}\right)^{51}+\left(1+\sqrt{5}\right)^{51}}{2\,251\,799\,813\,685\,248\sqrt{5}}$$

```
Formula /. n → 50 // Expand
```

20 365 011 074